




21世纪全国本科院校电气信息类**创新型**应用人才培养规划教材

微机原理及接口 技术实验指导书

主 编 李干林 李 升
主 审 韩念杭



北京大学出版社
PEKING UNIVERSITY PRESS

21 世纪全国本科院校电气信息类创新型应用人才培养规划教材

微机原理及接口技术实验指导书

李干林 李 升 主编
韩念杭 主审



北京大学出版社
PEKING UNIVERSITY PRESS

内 容 提 要

本书是配合微机原理及应用等课程的教学而编写的实验及课程设计指导书,书中以星研 STAR ES598PCI 实验系统为实验设备,较详细介绍了本课程所需的实验。实验内容紧跟理论教学进程,在接口及应用方面有较丰富的扩展。为方便学生的学习,实验备有参考程序,并补充有部分演示实验。同时,对接口实验需要的一些理论知识也进行了必要的补充。课程设计部分选择了四个较有代表性的课题。书后附有汇编语言上机参考、星研软件的使用介绍、实验设备的组成等,供使用者参考。

本书可作为本科院校工科类专业实验教材,也可作为工程应用人员的设计参考书。

图书在版编目(CIP)数据

微机原理及接口技术实验指导书/李干林,李升主编. —北京:北京大学出版社,2010.8

(21 世纪全国本科院校电气信息类创新型应用人才培养规划教材)

ISBN 978-7-301-17614-6

I. ①微… II. ①李…②李… III. ①微型计算机—理论—实验—高等学校—教学参考资料②微型计算机—接口—实验—高等学校—教学参考资料 IV. ①TP36-33

中国版本图书馆 CIP 数据核字(2010)第 151923 号

书 名: 微机原理及接口技术实验指导书

著作责任者: 李干林 李 升 主编

策 划 编 辑: 程志强

责 任 编 辑: 郑 双

标 准 书 号: ISBN 978-7-301-17614-6/TP·1124

出 版 者: 北京大学出版社

地 址: 北京市海淀区成府路 205 号 100871

网 址: <http://www.pup.cn> <http://www.pup6.com>

电 话: 邮购部 62752015 发行部 62750672 编辑部 62750667 出版部 62754962

电 子 邮 箱: pup_6@163.com

印 刷 者: 北京鑫海金澳胶印有限公司

发 行 者: 北京大学出版社

经 销 者: 新华书店

787 毫米×1092 毫米 16 开本 11 印张 246 千字

2010 年 8 月第 1 版 2010 年 8 月第 1 次印刷

定 价: 22.00 元

未经许可,不得以任何方式复制或抄袭本书之部分或全部内容。

版权所有,侵权必究

举报电话: 010-62752024

电子邮箱: fd@pup.pku.edu.cn

前 言

微机原理及应用是一门理论性和实践性很强的课程,是计算机、电力、自动化、通信工程、机电一体化等多个专业的专业基础课,是培养学生的微机软、硬件设计与应用能力、工程意识和创新能力的一门必修课。因此,要使学生掌握这门课程,必须在重视理论教学的同时,充分重视实验教学,并通过实验教学环节增强学生动手解决实际问题的能力。

本书根据编者多年的实践教学经验编写,在编排上紧跟理论教学进程,以针对性强的独立软硬件实验为基础,通过实验使学生掌握计算机软、硬件工作原理及常用接口芯片的功能和基本用法;在内容上主要安排了一些基础性实验,包括多个基础软件实验、基础硬件实验,例如并行/串行接口、可编程定时器/计数器、中断、A/D 转换、D/A 转换、8279 键盘显示等内容,以及综合应用实验。这些实验在相应章节的课堂教学结束时即进行上机调试、验证,从而可加深学生对所学知识的理解。每个实验均有几个验证性实验和程序设计题,学生通过阅读并调试验证性实验,可学会基本的编程方法,对接口电路有初步的了解,然后通过自己设计电路和编制程序,进一步掌握微机原理软件程序的编制及硬件接口电路的设计。本书安排了多个软件及硬件综合性实验,以帮助学生培养良好的软件及硬件的工程设计风格,并使其熟练地掌握软硬件的调试方法。

为配合微机原理及接口技术课程设计,使学生对所学的知识有更进一步的理解,并能把课本中分布学习的知识,在设计中综合加以应用,进而得到巩固和拓展,本书还设计了 4 个课程设计课题。通过课程设计可以使学生学习查找、运用设计资料的方法,并能掌握学习理论时没有注意的细节,完成工程设计必备的基本训练。

本书中的实验和课程设计均可在星研 STAR ES598PCI 实验系统上完成,也可以用于其他的实验系统(部分实验程序和硬件接线应加修改)。

本书由南京工程学院李干林、李升担任主编,李升编写了本书基础软件实验的实验一、基础硬件实验的实验二和附录 B,南京工程学院韩念杭编写了课程设计题目二、三并担任主审,其他部分由李干林编写。感谢上海星研电子科技有限公司提供的参考资料,也感谢南京工程学院鞠阳副教授的帮助!

限于能力和水平,本书中难免会存在不足与问题,殷切期盼能够得到广大读者的宝贵意见和建议。

编 者
2010 年 4 月

目 录

第 1 部分 基础软件实验	1	第 3 部分 综合实验	83
实验一 DEBUG 入门	1	实验一 简易电子琴实验	83
实验二 编辑、汇编、链接和 调试源程序	6	实验二 步进电动机实验	91
实验三 顺序结构程序设计	9	实验三 电子钟(CLOCK)	98
实验四 分支结构程序设计	11	第 4 部分 课程设计	105
实验五 循环结构程序设计	14	设计题目一 带时间显示的十字路口交通 信号模拟装置设计	105
实验六 子程序设计	16	设计题目二 交通信号灯模拟控制 系统的设计二	111
实验七 冒泡排序	18	设计题目三 全自动洗衣机微机 控制器设计	120
实验八 二分查找法	20	设计题目四 电子钟程序设计	123
第 2 部分 基础硬件实验	23	附录 A 汇编语言上机实验基础	126
实验一 8255A 并行输入/输出 接口实验	23	A.1 汇编语言程序上机实验过程	126
实验二 8255A 与键盘、LED、 数码管显示	30	A.2 宏汇编程序 MASM	127
实验三 8253 定时计数器实验	37	A.3 调试程序 DEBUG	129
实验四 8259A 中断控制器实验	40	A.4 调试程序 TDEBUG	136
实验五 8251A 可编程通信实验	49	A.5 EMU 8086	137
实验六 8251A 可编程通信实验 (与微机)	54	A.6 总结	139
实验七 并行 D/A 实验	58	A.7 实验及实验报告要求	139
实验八 并行 A/D 实验	59	附录 B 星研实验仪介绍	140
实验九 8250 可编程通信实验	62	B.1 功能特点	140
实验十 8279 键盘显示实验	66	B.2 软件	142
实验十一 8237 DMA 传输 实验	69	B.3 硬件	150
实验十二 图形点阵显示实验	73	参考文献	165

第 1 部分

基础软件实验

基础软件实验部分由 8 个实验组成，通过对这些实验中程序的编写、调试，使学生熟悉 8086/8088 的指令系统，了解程序设计过程，掌握汇编程序设计方法以及使用实验系统提供的各种调试、分析手段来排除程序错误的方法。

实验一 DEBUG 入门

一、实验目的和要求

1. 学习、了解 DEBUG 常用命令(A、U、D、E、R、T、G、Q)的使用方法。
2. 通过观察，了解部分数据传送类指令的功能和行为特征。
3. 了解 DEBUG 中 FLAGS 的各标志的表示方法。

二、实验任务

用 DEBUG 命令输入部分指令，执行并观察执行的结果。

三、实验设备和软件

PC、DEBUG.EXE。

四、预习内容

1. 调试程序 DEBUG 的常用命令(参考附录 A.3 小节)
2. DEBUG 中 FLAGS 的各标志的表示方法，见表 1-1。

表 1-1 状态标志显示符号

标志位	溢出 OF	方向 DF	中断 IF	符号 SF	零位 ZF	辅助 AF	奇偶 PF	进位 CF
状态	有/无	减/增	开/关	负/正	零/非	有/无	偶/奇	有/无
显示	OV/NV	DN/UP	EI/DI	NG/PL	ZR/NZ	AC/NA	PE/PO	CY/NC

3. 16 位标志寄存器 FLAGS 的格式见表 1-2。
4. 8086/8088 的各类指令。



表 1-2 16 位标志寄存器 FLAGS 的格式

D15	D14	D13	D12	D11	D10	D9	D8	D7	D6	D5	D4	D3	D2	D1	D0
				OF	DF	IF	TF	SF	ZF		AF		PF		CF

五、实验内容

1. 数据传送类指令观察

首先用 A 命令从地址 2000: 0000 开始依次逐条输入以下所有指令, 然后用 T 命令单步执行每一条指令, 其他要求请参见各条指令的注释部分。通过实验理解每条指令的功能。

指令助记符	操作数表	注释
MOV	AL, 78H	;用 R 命令显示 AL 的值
MOV	AX, 1234H	;用 R 命令显示 AX 的值
MOV	BX, 0200H	;用 R 命令显示 BX 的值
MOV	[BX], AX	;用 D 命令显示存储器单元中的值
MOV	SI, [0200H]	;用 R 命令显示 SI 的值
MOV	DH, AL	;用 R 命令显示 DH 的值
MOV	WORD PTR 2[BX], 5600H	;用 D 命令显示存储器单元中的值
MOV	CX, -1233H[BX+SI]	;用 R 命令显示 CX 的值
PUSH	AX	;用 D 命令显示堆栈情况
PUSH	2[BX]	;用 D 命令显示堆栈情况
PUSHF		;用 D 命令显示堆栈情况, 和 FLAGS 标志对比
POP	[BX]	;用 D 命令显示存储器单元中的值
POP	AX	;用 R 命令显示 AX 的值
POPF		;查看 FLAGS 各标志情况
XCHG	AX, DX	;用 R 命令显示 AX 和 DX
XCHG	DL, 2[BX]	;用 R 命令、D 命令分别查看 DL 和存储器单元
LEA	DX, -FF00H[BX+SI]	;查看 DX 的值
LAHF		;查看 AH 的值
NOT	AH	;查看 AH 的值
SAHF		;查看 FLAGS 低 8 位标志情况

在实验过程中记录指令执行的结果, 并在实验报告中以表格的形式体现。

2. 查表程序

首先用 E 命令从 DS: 0200H 开始的内存中写入一个 0~9 的平方表, 然后用 A 命令输入以下查表程序并执行, 观察查表结果。

```
MOV    AL, 5
MOV    BX, 0200H
XLAT
```

提示: (1) 输入的平方值表必须转换成十六进制数。

(2) R 命令只能查看 16 位寄存器的内容, 其中 16 位的前 8 位是高 8 位寄存器的内

容, 后 8 位是低 8 位寄存器的内容。

六、实验报告

实验报告的要求参考附录 A.7 小节, 下面给出中断控制器 8259A 实验的实验报告供参考。

1. 实验目的和要求

(1) 了解 8259A 的内部结构、工作原理, 掌握对 8259A 的初始化编程方法, 了解 8088 是如何响应中断、退出中断的。

(2) 了解 8259A 与 8088 的接口逻辑。

2. 实验仪器和设备

PC 一台, STAR 系列实验仪一套。

3. 实验内容

1) 实验题目

8259A 的 IR0 接单脉冲开关, 每次按动单脉冲开关使 8259A 响应外部中断 IR0 时, 实验中连接的 8 个发光二极管 LED 显示相对应拨动次数的 8 位二进制编码值, 如: 单脉冲开关第一次按动, 显示 00000001B, 只有 LED0 亮; 第二次按动单脉冲开关, 显示 00000010B, 只有 LED1 亮。

2) 实验设计思路

(1) 硬件设计。使用中断控制器 8259A 和并行接口 8255A, 其中 8259A 的 IR0 接单脉冲开关来实现中断功能, 8255A 输出显示中断次数, 可用 8255A 的端口 A(工作在方式 0 输出)接 8 位发光二极管显示。

(2) 程序设计。编制的程序主要由 8259A 和 8255A 的初始化程序、中断服务程序、中断向量写入中断向量表程序、中断等待主程序构成。8255A 初始化是确定工作方式, 8259A 的初始化是写入初始化命令字和工作方式命令字。中断服务程序的内容是从 8255A 的端口 A 输出显示的中断次数。中断等待主程序可以采用循环的方式等待中断。

3) 8259A 实验原理图

8259A 的实验原理图如图 1.1 所示。




图 1.1 8259A 实验原理图



4) 实验连线说明

B4 区: CS、A0、A1	—	A3 区: CS5、A0、A1
B4 区: JP56(PA 口)	—	G6 区: JP65
B3 区: CS、A0	—	A3 区: CS1、A0
B3 区: INT、INTA	—	ES8688: INTR、INTA
B3 区: IR0	—	B2 区: 单脉冲

5) 实验步骤

(1) 按连线说明在 STAR 系列实验仪上连接 8259A、8255A 芯片的导线。

(2) 编写程序并调试运行。

(3) 在全速运行程序下, 拨动单脉冲开关输入中断脉冲, 查看发光二极管的亮灭是否符合预期效果。

6) 实验程序编写

```

DATA          SEGMENT
IO8259A_0     EQU        0F000H
IO8259A_1     EQU        0F001H
K8255A        EQU        0B000H
K8255K        EQU        0B003H
BUFFER        DB         1,2,3,4,5,6,7,8,9      ;发光二极管显示的中断次数
DATA          ENDS
CODE          SEGMENT
              ASSUME     CS:CODE, DS:DATA
START:        MOV        AX, DATA
              MOV        DS, AX
              MOV        ES, AX
              NOP
              MOV        DX, K8255K
              MOV        AL, 80H
              OUT        DX, AL                  ;初始化 8255A 使 A 口为输出端口
              CALL       Init8259A               ;调用 8259A 初始化子程序
              CALL       WriIntver               ;调用写 8259A 中断服务程序的中断向量的子程序
              MOV        SI, OFFSET BUFFER
              MOV        BL, 1
              STI
              NOP
START1:        INC        BX
              JMP        START1                  ;循环等待中断
Init8259A     PROC        NEAR
              MOV        DX, IO8259A_0

```



```

MOV     AL,13H
OUT     DX,AL                ;写 8259A 初始化字 ICW1
MOV     DX,IO8259A_1
MOV     AL,08H
OUT     DX,AL                ;写 8259A 初始化字 ICW2
MOV     AL,09H
OUT     DX,AL                ;写 8259A 初始化字 ICW4
MOV     AL,0FEH
OUT     DX,AL                ;写 8259A 操作字 OCW1
RET

Init8259A                     ENDP
WriIntver PROC NEAR
PUSH    ES
MOV     AX,0
MOV     ES,AX
MOV     DI,20H
LEA     AX,INT_0
STOSW                      ;写入中断向量的 IP 值
MOV     AX,CS
STOSW                      ;写入中断向量的 CS 值
POP     ES
RET
WriIntver ENDP
INT_0:  PUSH    DX                ;中断服务程序
        PUSH    AX
        MOV     DX,K8255A
        MOV     AL,[SI]
        NOT     AL
        OUT     DX,AL            ;显示中断次数
        INC     SI
        MOV     DX,IO8259A_0
        MOV     AL,20H
        OUT     DX,AL            ;写 OCW2 命令,即发 EOI 命令
        POP     AX
        POP     DX
        IRET                     ;中断返回
CODE    ENDS
END      START

```

4. 实验分析

(1) 通过观察实验,当拨动单稳态脉冲的开关,相应的发光二极管显示内容是拨动开关的次数,验证了中断控制器 8259A 的中断管理功能。

(2) 程序数据段中只放置了 1~9 的数字,当拨动开关的次数超过 9 次时,显示的内



容就是不确定了。这说明程序各段之间是相互关联的。如果要一直显示正确次数，可按如下方法修改代码。

在主程序中：

```
MOV    CL,1    ;初始化显示内容
```

在中断服务程序中：

```
MOV     DX,K8255A
MOV     AL,CL
OUT     DX,AL    ;显示中断次数
INC     CL
```

实验二 编辑、汇编、链接和调试源程序

一、实验目的和要求

1. 掌握汇编语言程序编辑、汇编、链接和调试的基本概念。
2. 掌握在 PC 上建立、汇编、链接和运行 8088 汇编语言程序的操作方法。
3. 巩固 DEBUG 常用命令的使用方法。
4. 通过观察，深入掌握数据传送类、算术运算类和逻辑运算类等各类指令的功能和行为特征。
5. 在 DEBUG 中观察指令对各标志位的影响，以及标志状态对部分指令的影响。

二、实验任务

编辑汇编语言源程序并汇编、链接和调试，验证结果。

三、实验设备和软件

PC、MASM. EXE、LINK. EXE、DEBUG. EXE。

四、预习内容

1. 预习附录 A 中的汇编语言程序上机实验过程、宏汇编程序 MASM、调试程序 DEBUG 内容。
2. 预习实验内容的程序，通过实验验证结果。

五、实验内容

1. 寻找最大、最小数的程序实验

设内存缓冲区从 BUF 单元开始存放若干个单字节数，其数据长度为 B1，要求找出最大数送 MAX 单元，最小数送 MIN 单元。

题目分析：可先将第一个数据赋值给最大数和最小数，后面的数据和最大数、最小数比较，大于最大数就与最大数换值，小于最小数就与最小数换值，直至所有数据比较



完毕。

程序流程图如图 1.2 所示。




图 1.2 程序流程图

源程序如下。

```

;FILENAME:MAXMIN.ASM
标号/名字  指令/伪指令  操作数表  ;注释
DATA      SEGMENT
BUF        DB          9,-4,55,78,-9,0,41,124
B1         EQU         $-BUF
MAX        DB
MIN        DB
DATA      ENDS
CODE      SEGMENT
          ASSUME      CS:CODE,DS:DATA

```

```

START:  MOV     AX, DATA
        MOV     DS, AX
        MOV     CH, 0
        MOV     CL, B1           ;CX=数据个数
        DEC     CX               ;循环次数
        MOV     AL, BUF         ;第一个数据送 AL 寄存器
        MOV     MAX, AL         ;假设第一个数是最大数
        MOV     MIN, AL         ;假设第一个数是最小数
        MOV     BX, OFFSET BUF+1 ;下一个数据地址
LAST:   MOV     AL, [BX]
        CMP     AL, MAX         ;比较,注意标志变化
        JG      GREAT          ;注意标志对指令行为的影响
        CMP     AL, MIN         ;比较,注意标志变化
        JL      LESS           ;注意标志对指令行为的影响
        JMP     NEXT
GREAT:  MOV     MAX, AL         ;大数→MAX
        JMP     NEXT
LESS:   MOV     MIN, AL         ;小数→MIN
NEXT:   INC     BX
        LOOP    LAST
        NOP                    ;实验时程序执行到此,通过 D 命令可查看数据段结果
        MOV     AH, 4CH
        INT     21H
CODE    ENDS
        END     START

```

2. 代码转换实验

从键盘输入两个十进制数并组合成压缩 BCD 码存入 DL, 再将压缩 BCD 码转换成 ASCII 码送 CRT 显示。

源程序如下。

```

CODE    SEGMENT
        ASSUME  CS:CODE
ST:      MOV     AH, 01H         ;键盘输入并回显,AL=输入字符
        INT     21H
        MOV     CL, 4
        SHL     AL, CL          ;注意标志变化
        MOV     DL, AL
        MOV     AH, 01H
        INT     21H
        AND     AL, 0FH         ;注意标志变化
        OR      DL, AL          ;注意标志变化
        MOV     BL, DL          ;保护 DL 原值

```



```

SHR      DL, CL          ;注意标志变化
OR       DL, 30H         ;注意标志变化
MOV      AH, 02H         ;显示十位数, DL=输出字符
INT      21H
MOV      DL, BL
AND      DL, 0FH         ;注意标志变化
OR       DL, 30H
MOV      AH, 02H         ;显示个位数
INT      21H
MOV      AH, 4CH
INT      21H
CODE     ENDS
END      ST

```

要求先建立文件夹, 用 EDIT. EXE 编写汇编文件, 运用 MASM. EXE、LINK. EXE 进行汇编和链接, 运用 DEBUG 或 EMU8086 采用单步调试的方法进行运行调试, 任意输入两个数字, 验证实验结果。

六、实验报告

实验三 顺序结构程序设计

一、实验目的和要求

1. 熟悉汇编语言集成编译软件的使用方法。
2. 熟悉 8086 汇编指令, 能自己编写简单的程序并调试运行验证结果。

二、实验任务

学习使用汇编语言集成编译软件, 软件提供了编辑、编译、运行、调试汇编语言程序的集成环境。验证及编写、调试实验内容。

三、实验设备和软件

汇编语言集成编译软件(如星研集成环境软件, 见附录 B)。

四、预习内容

1. 熟悉汇编语言集成编译软件的使用。
2. 复习顺序结构程序的设计方法, 预习实验内容。

五、实验内容

1. 编写程序, 计算 $[W - (X + 120) * Y + 200] / Z$ 的值, 其中 W、X、Y、Z 均为 16 位的带符号数。将计算结果的商和余数分别存放在以 RESULT 单元开始的数据区中。



程序清单如下。

```
DATA    SEGMENT
W        DW      1200          ;数据定义
X        DW      -30
Y        DW      -25
Z        DW      500
RESULT   DW      2 DUP (?)
DATA     ENDS
CODE     SEGMENT
        ASSUME  CS:CODE, DS:DATA
BG:      MOV     AX, DATA
        MOV     DS, AX
        MOV     AX, X
        ADD     AX, 120        ;X+120, 注意标志变化
        IMUL    Y              ;(X+120)*Y, 注意标志变化
        MOV     CX, AX
        MOV     BX, DX        ;乘积暂存 BX: CX
        MOV     AX, W         ;将 W 带符号扩展
        CWD
        SUB     AX, CX        ;实现 W-(X+120)*Y
        SBB     DX, BX        ;结果在 DX:AX 中, 注意标志变化
        ADD     AX, 200       ;实现 W-(X+120)*Y+120, 注意标志变化
        ADC     DX, 0         ;结果在 DX:AX 中, 注意标志变化
        IDIV    Z            ;最后除以 Z, 商在 AX 中, 余数在 DX 中, 注意标志变化
        MOV     RESULT, AX    ;存放结果到数据区
        MOV     RESULT+2, DX
        MOV     AH, 4CH       ;返回到 DOS
        INT     21H
CODE     ENDS
        END     BG           ;汇编结束
```

2. 编写程序, 实现组合 BCD 码转换成 ASCII 码并送 CRT 显示。

程序清单如下。

```
CODE     SEGMENT
        ASSUME  CS:CODE
BEGIN:   MOV     AL, 27H      ;27 的 BCD 码, 先显示高位
        MOV     DL, AL
        MOV     CL, 4
        SHR     DL, CL
        OR      DL, 30H
        MOV     AH, 2
```




```

INT      21H      ;显示"2"
MOV      DL,AL
AND      DL,0FH
OR       DL,30H
MOV      AH,2
INT      21H      ;显示"7"
MOV      AH,4CH
INT      21H
CODE     ENDS
END      BEGIN

```

3. 试编写：设 x 、 y 、 z 已定义为字节变量，若 x 和 y 各存放一个 32 位(四字节)的无符号数，存放顺序是低位字节在先，试写出将 x 和 y 相加、结果存入 z 的程序段。

4. 编写程序实现运算(式中 $W1$ 、 $W2$ 、 $W3$ 、 $W4$ 为字变量，运算结果的商存放在 $W3$ ，余数存放在 $W4$ 中)： $(W1 - W2) / 10 \rightarrow W3, W4$ 。

六、实验报告

实验四 分支结构程序设计

一、实验目的和要求

1. 学习分支结构程序的设计和调试方法。
2. 熟悉 8086 汇编指令中的程序控制指令的应用。

二、实验任务

学习使用汇编语言集成编译软件，验证及编写、调试实验内容。

三、实验设备和软件

汇编语言集成编译软件。

四、预习内容

1. 熟悉汇编语言集成编译软件的使用。
2. 复习分支结构程序的设计方法，预习实验内容。

五、实验内容

1. 已知在内存中有一个字节单元 NUM ，存有带符号数据，要求计算出它的绝对值后放入 $RESULT$ 单元中。

题目分析：由数学中绝对值的概念知道，一个正数的绝对值是它本身，而一个负数的绝对值是它的相反数；要计算一个数的相反数，需要完成减法运算，即用 0 减去这个

数。8086/8088 指令系统中有专门的求相反数的指令 NEG。
程序流程图如图 1.3 所示。




图 1.3 程序流程图

程序清单如下。

```

DATA    SEGMENT
X        DB      -25
RESULT   DB
DATA     ENDS
CODE     SEGMENT
        ASSUME   CS:CODE, DS:DATA
START:   MOV     AX, DATA
        MOV     DS, AX                ;初始化
        MOV     AL, X                ;X 取到 AL 中
        TEST    AL, 80H              ;测试 AL 正负
        JZ      NEXT                ;为正, 转 NEXT
        NEG     AL                    ;否则 AL 求补
NEXT:    MOV     RESULT, AL           ;送结果
        MOV     AH, 4CH
        INT     21H                  ;返回 DOS
CODE     ENDS
        END      START                ;汇编结束
  
```

2. 编制程序将一串字节符号数中的正、负数分别送到 PLUS 和 MINUS 为首地址的存储区中去, 同时记录 0 的个数(小于 65 536)。

程序流程图如图 1.4 所示。